常用算法BASIC语言应用手册

高崇　罗忠禹　王树清　索素文　高兴　编著

BASIC

中国铁道出版社

常用算法BASIC语言应用手册

中国铁道出版社

常用算法 BASIC 语言应用手册

高崇　罗忠禹　王树清　索素文　高兴　编著

中　国　铁　道　出　版　社
1 9 9 3 年·北京

(京)新登字 063 号

内 容 简 介

本书是计算方法和程序设计语言应用的工具书。其特点是内容丰富、通用性强、使用方便及可靠性高。其宗旨是实用。

本书共分两篇。第一篇为程序篇,第二篇为应用篇。其内容包括:插值与微商;数值积分;线性代数计算;非线性方程(组)的计算;常微分方程(组)的计算;最优化;拟合与平滑;数据处理与回归分析;特殊函数;快速付里叶变换;概率统计;等等。

为方便读者,书中每个程序前均有其功能及特点介绍、数学方法简介、程序说明及程序附注等。本书的应用篇列举了大量例子,以此帮助读者掌握和运用 BASIC 语言编制程序,解决工作中的实际问题。

本书可供工程技术人员及有关专业人员作为手册查阅,也可供理、工科院校非计算机专业师生学习参考。

常用算法 BASIC 语言应用手册

高崇 罗忠禹 王树清 索素文 高兴 编著

*

中国铁道出版社出版、发行

(北京市东单三条 14 号)

责任编辑 郭宇 封面设计 陈东山

各 地 新 华 书 店 经 售

中国铁道出版社印刷厂印

开本:787×1092 毫米 1/16 印张:22.5 字数:557 千

1993 年 9 月 第 1 版 第 1 次印刷

印数:1—2500 册

ISBN7-113-01405-4/TP·137 定价:14.35 元

前　　言

目前，计算机的应用越来越普遍，尤其是微机，不论是科研单位、大专院校、还是工矿企业，其数量已相当可观。在这些计算机中，用户使用编程最为广泛的语言就是BASIC语言，然而目前已出版的用BASIC语言编写的程序书籍往往还不能满足人们的需要。为了使广大科学工作者、工程技术人员、大专院校师生及管理工作者能更有效地利用计算机解决工作中遇到的问题，减少不必要的重复劳动，我们编写了此书，希望它能给广大使用者带来方便。

与其他的BASIC程序书籍相比，本书具有以下几个特点：

(1)内容十分丰富。本书共有160个程序，几乎包括所有的供一般科学及工程计算用的常用算法。对每类问题我们都编写了多个程序，每个程序又各具特色，因此可以有效地启发读者，帮助读者解决工作中存在的问题。

(2)通用性强。考虑到中、低档微机对变量、数组及字符长度的限制，本书中的程序只使用不超过两个字符的变量或数组，而且程序中只使用标准的、基本的BASIC语句，因而这些程序对使用BASIC语言的所有高、中、低档计算机用户均适用，一般不需再作任何修改。

(3)使用方便。本书中的程序一般只需作少量的输入便可运算。并且，书中的程序均以子程序的形式给出，使用者只需按程序说明写出输入输出语句即可，而无需对程序进行修改。另外，使用苹果机、IBMPC及兼容机的用户可以直接应用本书配套提供的软磁盘程序实例来解决工作和学习中一些需要编程的问题。由于苹果机及兼容机软磁盘内具有更改行号、程序联接软件，使用者可利用它将软磁盘中的程序任意组合。

(4)可靠性高。绝大多数程序均有计算标志，使用者可根据其值来判断计算结果的优劣。

(5)本书附有较多的应用实例，使用者可从中看出程序的使用方法及对各类数学问题的处理技巧。这些内容许多是其他同类书中所没有的。

本书的宗旨是实用。所以每个程序(算法)均按下面的统一格式编写：

一、功能及特点

简要地说明程序的用途、适用范围及特点。这样即使对算法本身不太了解的人也可以应用。

二、方法简介

考虑到许多使用者对常用算法都有一定的了解，他们所希望的是怎样更好地编写程序。所以本书的重点是如何引导读者编程，而对较常用的算法书中仅给出参考文献。对于一些不经常用的算法我们附以简明的介绍。

三、程序说明

1. 程序名称

2. 输入参数

指明调用程序时必须输入的变量和数组等的意义，类型和数量等，使用者只需按此说明输入即可执行运算。

3. 输出参数

指明程序执行后可输出(有意义)的变量、数组的意义、类型和数量等，一般情况下使用者

只要照此说明输出即可。

4. 所调用的子程序

指出调用该程序时，还需要调用哪些子程序，并指明是否由使用者自编。如果需要使用者自编程序，书中对变量和函数等分别加以说明。

四、程序附注

给出对程序的附加说明。一般情况下程序本身是不用修改的，当遇到某些特殊情况时，使用者可照此说明对程序作适当的修改。

五、程序

书中的程序均以 BASIC 语言编写。

本书共分两篇。第一篇为程序篇，第二篇为应用篇。第一篇中每章开头部分都有一个应用指南性质的介绍，重点介绍各个算法的功用、特点及适用范围等，以便于读者正确地选用程序。第二篇中列举了大量的例题，考虑到通用性，例题均以数学形式给出。编写此篇的目的主要有两点：

(1)作为第一篇各章的例题，使读者能更清楚地了解本书中各个程序的应用范围、特点及调用方法，以便读者能更好地利用这些程序解决实际问题；

(2)通过计算实例介绍，对某些问题给出处理方法或技巧。

实际的计算问题往往是很复杂的，因此许多问题不能直接应用书中的程序求解或计算，但采用某些处理方法或技巧后，就会将原来很复杂的问题分解成若干个较容易处理的子问题或将原来不能直接利用程序计算的问题化为经若干程序组合后便可计算的问题。

例题虽然较多，但仍有相当数量的程序没举例子，因为许多程序的调用方法是很类似的。例题中调用了许多子程序，书中只给出子程序的名称，读者可从《常用算法程序》软磁盘中查出这些子程序在书中第一篇中的位置。

第一篇的内容均存于《常用算法程序》软磁盘中，由于受软磁盘目录区及盘片数目的限制，所以本书中的部分程序是以两个程序合并应用一个程序名存盘。例如：(1)程序名为 CZ1—CZ2 表示程序 CZ1 与 CZ2 合并后存盘，其中 CZ1 的行号是 50 开始的，CZ2 的行号是从 500 开始；(2)程序名为 FXXFC2—3 则表示程序 FXXFC2 与 FXXFC3 合并后存盘，其中 FXXFC2 的行号从 50 开始，FXXFC3 的行号从 500 开始，如调用子程序，可在相应的行号位置填写其内容。书中每个程序除 50 语句外，其余语句 REM 中的内容均应写入主程序中。

第二篇中所有例题的计算程序均存于《应用实例程序》软磁盘中，读者可在相应的微机上执行运算，检验计算结果。如读者改变数据或计算式，便可用来解决实际问题。由于第二篇中的计算程序也很丰富且涉及面较广，所以读者利用这些程序可解决工作中的许多计算问题。

本书中所有程序均上机通过，许多程序经多次计算验证是十分有效的。但由于我们水平有限，错误难免，欢迎给予批评、指教。

书中参考了许多文献，谨向文献作者们表示最诚挚的谢意。

编　者

一九九一年十一月

目　录

第一篇　程序及使用说明

第二篇　应用实例

第一篇　程序及使用说明

第一章　插值与微商

利用已知数据进行插值、微商是工作中经常遇到的问题之一。本章的程序可以解决以下几类问题：

(1)求函数值；

(2)求一阶或二阶导数值；

(3)求区间内的积分值。

根据插值函数的构造不同，可以将其分为代数插值、有理插值和样条插值。这几种插值都有广泛的使用价值，但其特点又各不相同。

代数插值(多项式插值)公式由于运算简单，且在整个数轴上都有任意阶导数，故成为逼近光滑函数的极重要工具。在许多情况下，适当提高插值公式的次数便可以提高插值精度，但不宜过高。因为次数过高容易产生 Runge 现象——振荡，在振荡区域内插值效果极差。另外，对于下列情况代数插值也不适用：当函数 $f(x)$ 在某点 a 附近无界或当 $x\to\infty$ 而 $f(x)\to$ 定值。道理很简单，因为多项式是反映不出 $f(x)$ 这两个特性的($x\to\infty$ 时，$f(x)\to\infty$；$x\to a$ 时，$f(x)$ 不可能无界)。

有理插值计算比较复杂，但对上述情况却很有效。插值结果与插值点数关系不十分紧密，增加插值节点，一般都可取得良好的结果；不易产生振荡现象，适用于绝大多数插值问题。另外对函数在某个区域内变化剧烈或变化十分平缓的问题也适用。其缺点就是计算速度稍慢，但当插值点不是很多时并不十分明显。

样条插值是很受重视的一种方法，尤其是三次样条。样条插值不但能保证函数值的连续性，还有保证其光滑性，所以更适合于作光滑插值，当然对于一般的插值也很有效。

有些函数在不同区域内有不同的特性，这就需要分段插值。本章中的差商检验是为考察分段情况编写的，从各阶差商值可以大致判断出函数值的变化范围，从而确定分段数及分段区间。另外，该程序还为确定代数插值的阶数提供了一定的依据，使用者可以方便地判断出采用几个点进行插值为“最佳”。

本章中的程序有的可以进行外插，有的只适用于内插，但外插时一定要慎重。一般情况下内插总是要好于外插。函数变化比较均匀的一般不用分段。若函数确需分段插值，一定要正确选择分段区域，否则仍得不到理想的插值结果。

本章中第七至十一个程序不但能进行插值，还能求微商和积分值，其中 CZ7 还能求区间内任意两点间的积分值。

本章还有一个利用外推法计算一阶导数和利用中心差分法计算一、二阶导数的程序。如果函数表达式已知但却很复杂且不易求导时，采用这两个程序是很适用的，它们的特点是快速而又准确。

第一节　一元 M 点不等距插值

一、功能及特点

采用拉格朗日公式从 N 个插值点中选取最靠近插值点的相邻 M 个插值节点进行插值。对等距情况也适用。

二、方法简介

详见参考文献[1,3,7]。

三、程序说明

1. 程序名称：CZ1

2. 输入参数

N——插值节点个数；

M——选用的插值节点个数；

XT——插值点；

X0——N 个元素的一维数组，存放给定的插值节点，要求 $X0_i < X0_{i+1}$；

Y0——N 个元素的一维数组，存放节点上的函数值。

3. 输出参数

Y——插值结果。

四、程序附注

M 值不宜选得过大，一般 M≤8。

五、程　　序

```
CZ1:
50  REM YI YUAN M DIAN BU DENG JU CHA ZHI
55  REM X(M),Y(M),X0(N),Y0(N)
60  W=INT(M/2)
65  FOR I=1 TO N-W
70  IF XT>X0(I) GOTO 115
75  IF I<M-1 GOTO 100
80  II=I
85  FOR J=1 TO M
90  Y(J)=Y0(II-J-W-1): X(J)=X0(II+J-W-1)
95  NEXT J:GOTO 135
100 FOR JJ=1 TO M
105 Y(JJ)=Y0(JJ):X(JJ)=X0(JJ)
110 NEXT JJ:GOTO 135
115 NEXT I
120 FOR I=1 TO M
125 X(I)=X0(N-M+I):Y(I)=Y0(N—M+I)
130 NEXT I
135 Y=0
140 FOR I=1 TO M:P=1
145 FOR J=1 TO M
```

```
150  IF I<>J THEN P=P * (XT-X(J))/(X(I)—X(J))
155  NEXT J
160  Y=Y+P * Y(I)
165  NEXT I
170  RETURN
```

第二节　一元三点不等距成组插值

一、功能及特点

采用拉格朗日三点插值公式从 N 个给定的插值节点中选取最靠近插值点的相邻三个插值节点，对 L 个一元函数进行成组插值。

二、方法简介

详见参考文献[1,2,3]。

三、程序说明

1. 程序名称：CZ2

2. 输入参数

N——插值节点个数，N≥3；

L——插值的函数个数；

XT——插值点；

X——N 个元素的一维数组，存放给定的插值节点，要求 $X_i<X_{i+1}$；

Y——N * L 个元素的二维数组，存放节点上的函数值，存放顺序如下：$Y^{(1)}_{(X_1)}$，$Y^{(1)}_{(X_2)}\cdots$，$Y^{(1)}_{(X_n)}$，$Y^{(2)}_{(X_1)}$，$Y^{(2)}_{(X_2)}\cdots$，$Y^{(2)}_{(X_n)}$，…，$Y^{(L)}_{(X_1)}$，…，$Y^{(L)}_{(X_n)}$。

3. 输出参数

YY——L 个元素的一维数组，存放 L 个插值结果。

四、程　　序

CZ2：

```
50  REM YI YUAN SAN DIAN BU DENG JU CHENG ZU CHA ZHI
55  REM X(N),Y(N,L),YY(L)
60  N1=N—2
65  FOR I=1 TO N1
70  IF XT<=X(I+1) GOTO 85
75  NEXT I
80  I=N-2
85  IF I=1 GOTO 95
90  IF (XT-X(I))<(X(I+1)-XT) THEN I=I-1
95  A1=X(I):A2=X(I+1):A3=X(I+2)
100 U=(XT-A2) * (XT-A3)/(A1-A2)/(A1-A3)
105 V=(XT-A1) * (XT-A3)/((A2-A1) * (A2-A3))
110 W=(XT-A1) * (XT-A2)/((A3-A1) * (A3-A2))
115 FOR J=1 TO L
120 YY(J)=U * Y(I,J)+V * Y(I+1,J)+W * Y(I-2,J)
125 NEXT J
```

```
130 RETURN
```

第三节　Aitken 插值

一、功能及特点

采用 Aitken 反复线性插值公式从给定的 N 个插值节点中选取最靠近插值点的相邻 M 个插值节点对一元函数进行插值。

二、方法简介

详见参考文献[1,3,7]。

三、程序说明

1. 程序名称：CZ3

2. 输入参数

N——插值节点个数；

M——选用的插值节点个数；

XT——插值点；

X——N 个元素的一维数组，存放给定的插值节点，要求 $X_i < X_{i+1}$；

Y——N 个元素的一维数组，存放节点上的函数值。

3. 输出参数

Y——插值结果。

四、程序附注

M 不宜选得过大，一般 M≤8。

五、程　　序

```
CZ3:
50  REM AI TE JIN CHA ZHI
55  REM F(N),X(N),Y(N),Z(N)
60  EP=.000001
65  IF M>N THEN M=N
70  FOR I=1 TO N
75  IF XT<=X(I) GOTO 85
80  NEXT I:I=N
85  IF XT=X(I) THEN F(M)=Y(I):GOTO 170
90  IF INT(M/2)*2=M GOTO 105
95  IF I=1 GOTO 105
100 IF (XT-X(I-1))<(X(I)-XT) THEN I=I-1
105 I=I-M/2
110 IF I<=0 THEN I=1:GOTO 120
115 IF (I+M)>N THEN I=N-M+1
120 FOR J=1 TO M
125 Z(J)=XT-X(I):F(J)=Y(I):I=I+1
130 NEXT J:M1=M-1
135 FOR I=1 TO M1
140 FI=F(I):ZI=Z(I):I1=I+1
```

```
145 FOR J=I1 TO M
150 IF ABS(F(J)-F(J-1))<=EP THEN M=J:GOTO 170
155 F(J)=FI+ZI*(F(J)-FI)/(ZI-Z(J))
160 NEXT J
165 NEXT I
170 Y=F(M):RETURN
```

第四节 Hermite 插值

一、功能及特点

采用埃尔米特插值公式对一元函数进行插值。需已知节点上的导数值。

二、方法简介

详见参考文献[1,3,7]。

三、程序说明

1. 程序名称:CZ4

2. 输入参数

N——插值节点个数;

M——选用的插值节点个数,M≤N;

XT——插值点;

X0——N 个元素的一维数组,存放给定的插值节点,要求 $X0_i<X0_{i+1}$;

Y0——N 个元素的一维数组,存放节点上的函数值;

YY——N 个元素的一维数组,存放节点上的一阶导数值。

3. 输出参数

Y——插值结果。

四、程序附注

M 值不宜选得过大,一般 M=2～3。

五、程　　序

CZ4:

```
50 REM AI ER MI TE CHA ZHI
55 REM X(M),Y(M),X0(N),Y0(N),Y1(M),YY(N)
60 W=INT(M/2)
65 FOR I=1 TO N-W
70 IF XT>X0(I) GOTO 120
75 IF I<M-1 GOTO 105
80 II=I
85 FOR J=1 TO M
90 Y(J)=Y0(II+J-W-1):X(J)=X0(II+J-W-1)
95 Y1(J)=YY(II+J-W-1)
100 NEXT J:GOTO 140
105 FOR JJ=1 TO M
110 Y(JJ)=Y0(JJ):X(JJ)=X0(JJ):Y1(JJ)=YY(JJ)
115 NEXT JJ:GOTO 140
```